Annotating on Maps

**Purpose:** This chapter demonstrates how to annotate markers and paths on maps.

---

**Annotating Simple Point Markers**

Many times, all you want to do is place simple markers at specific locations on a map. You’re not trying to represent any data with the size/color/shape of the markers – you just want to mark some locations. Let’s start with this simple case, and later we’ll work our way up to more complex variations.

For example, let’s place markers on a map for a few of the cities in Texas. The following code will get the latitude and longitude (lat/long) for the desired cities from one of the datasets that come included with SAS/Graph. Of course, you don’t have to get the lat/long from this file – you could get the coordinates any way you want and store them in a similar dataset (look them up in a book, get them from a GPS, do a Google search, etc).

```sas
data anno_cities; set mapsgfk.uscity (where=(statecode='TX' and city in ('Houston' 'Dallas' 'Lubbock' 'San Antonio'))); flag=1;
run;
```

The resulting dataset looks like this:

<table>
<thead>
<tr>
<th>CITY</th>
<th>LAT</th>
<th>LONG</th>
<th>flag</th>
</tr>
</thead>
<tbody>
<tr>
<td>Dallas</td>
<td>32.7833</td>
<td>-96.800</td>
<td>1</td>
</tr>
<tr>
<td>Houston</td>
<td>29.7631</td>
<td>-95.383</td>
<td>1</td>
</tr>
<tr>
<td>Lubbock</td>
<td>33.5778</td>
<td>-101.855</td>
<td>1</td>
</tr>
<tr>
<td>San Antonio</td>
<td>29.4239</td>
<td>-98.493</td>
<td>1</td>
</tr>
</tbody>
</table>
Next, you’ll need the map of Texas. When plotting lat/long values on a map, I recommend using maps from the mapsgfk libname (in this case mapsgfk.us_states) rather than the traditional/old SAS maps such as maps.states. The reason being that your city coordinates are likely to be in eastlong degrees, and the mapsgfk maps are already in these coordinates (whereas the old/traditional SAS map coordinates are in westlong radians, and you would either have to convert your map or your city coordinates so that they match).

Here’s the code to subset Texas out of mapsgfk.us_states (using the lower density values will create a map that shows slightly less detail around the borders, and the code will also run a little faster):

```sas
data my_map; set mapsgfk.us_states
  (where=(statecode='TX' and density<=3));
run;
```

Our end goal is to plot markers at the city coordinates on the map. But first, it might be useful to see everything (the map border coordinates and the city coordinates) in a scatter plot, just so you can get a better feel for the data, and make sure that your city coordinates are lining up with your map coordinates. Here’s the code to do that (a little ‘fancier’ than necessary, but I wanted the graph to look pretty for the book!)

```sas
data combined; set my_map anno_cities; run;
symbol1 value=dot height=1.8 interpol=none color=gray55;
symbol2 value=dot height=3 interpol=none color=red;
axis1 major=(number=5) minor=none value=(c=blue)
  offset=(0,0);
proc gplot data=combined;
label lat='lat' long='long';
plot lat*long=flag / nolegend
  vaxis=axis1 haxis=axis1 noframe
  autohref autovref chref=graydd cvref=graydd;
run;
```
The next step is to project the map and city lat/long coordinates, to get the projected y & x coordinates you will plot with Proc GMap. In the case of Texas, projecting the coordinates doesn’t make much visual difference in the shape of the map, but the farther north you go, the more important projecting the coordinates becomes. Whether the map visually needs projecting, it is the ‘proper’ thing to do, and I recommend you always go through this step.

There are a couple of different ways you could project the coordinates, but the one I use most frequently is to combine the map and the annotate data, project them together, and then separate them into two datasets again. Here’s the code to do that (notice that I’m using the flag variable to separate the two datasets – this is the main purpose of having the flag variable). The GProject options signify that the coordinates are in eastlong degrees (as opposed to westlong radians), and stored in the lat and long variables (as opposed to y and x).

data combined; set my_map anno_cities; run;
proc gproject data=combined out=combined
eastlong degrees latlong dupok;
id statecode;
run;
data anno_cities my_map; set combined;
if flag=1 then output anno_cities;
else output my_map;
run;

The resulting datasets contain projected values, stored in the X and Y variables. Here’s a scatter plot so you can easily see what the data looks like:

The map dataset is ready to plot with Proc GMap, but the annotate dataset (for the cities) still needs a little work. Currently, it has the following variables that we’ll be using:
In order to annotate markers at those x/y coordinates, we’ll need to add a few variables to the dataset so that annotate will know what to do. The xsys and ysys tell annotate that the X and Y variables are in the same coordinate system as the map. The hsys tells it how to handle the size variable. When='a’ tells it to draw the annotation after it draws the map (so it will be in front of the map, rather than behind it). And the rest of the variables tell it to draw a solid red pie at the specified x/y location.

```sas
data anno_cities; set anno_cities;
length function $8;
xsys='2'; ysys='2'; hsys='3'; when='a';
function='pie'; style='psolid'; rotate=360; size=1;
  color='red';
run;
```

Here’s are the essential variables of the completed annotate dataset:
Now we can draw the map (with Proc GMap) and tell it to use anno_cities as the annotate dataset, with the following code:

```sas
title1 ls=1.5 "Some Cities in Texas";
pattern1 v=s c=white;
proc gmap map=my_map data=my_map anno=anno_cities;
  id statecode;
  choro segment / levels=1 nolegend coutline=gray88;
run;
```

Some Cities in Texas

Annotated Markers at City lat / long
One simple enhancement that is often useful, is to annotate a text label in addition to the pie marker. You can do this using the ... you guessed it! ... the label annotate function!

The following code annotates the city with each marker. The position='2' causes the label to be above the marker, and I reset the style/rotate/size/color variables so the default text characteristics will be used.

```plaintext
data anno_cities; set anno_cities;
  length function $8 text $100;
  xsys='2'; ysys='2'; hsys='3'; when='a';
  function='pie'; style='psolid'; rotate=360; size=1;
  color='red'; output;
  function='label'; position='2';
  style=''; rotate=.; size=.; color='';
  text=trim(left(city)); output;
run;
```

Here’s the resulting dataset:

```
<table>
<thead>
<tr>
<th>CITY</th>
<th>X</th>
<th>Y</th>
<th>xsys</th>
<th>ysys</th>
<th>hsys</th>
<th>when</th>
<th>function</th>
<th>style</th>
<th>rotate</th>
<th>size</th>
<th>color</th>
</tr>
</thead>
<tbody>
<tr>
<td>Dallas</td>
<td>0.049</td>
<td>0.029</td>
<td>2</td>
<td>2</td>
<td>3</td>
<td>'a'</td>
<td>'pie'</td>
<td>'psolid'</td>
<td>300</td>
<td>1</td>
<td>'red'</td>
</tr>
<tr>
<td>Dallas</td>
<td>0.048</td>
<td>0.029</td>
<td>2</td>
<td>2</td>
<td>3</td>
<td>'a'</td>
<td>'label'</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Houston</td>
<td>-0.023</td>
<td>-0.023</td>
<td>2</td>
<td>2</td>
<td>3</td>
<td>'a'</td>
<td>'pie'</td>
<td>'psolid'</td>
<td>300</td>
<td>1</td>
<td>'red'</td>
</tr>
<tr>
<td>Houston</td>
<td>-0.071</td>
<td>-0.023</td>
<td>2</td>
<td>2</td>
<td>3</td>
<td>'a'</td>
<td>'label'</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Lubbock</td>
<td>-0.028</td>
<td>0.042</td>
<td>2</td>
<td>2</td>
<td>3</td>
<td>'a'</td>
<td>'pie'</td>
<td>'psolid'</td>
<td>300</td>
<td>1</td>
<td>'red'</td>
</tr>
<tr>
<td>Lubbock</td>
<td>-0.026</td>
<td>0.042</td>
<td>2</td>
<td>2</td>
<td>3</td>
<td>'a'</td>
<td>'label'</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>San Antonio</td>
<td>0.024</td>
<td>-0.030</td>
<td>2</td>
<td>2</td>
<td>3</td>
<td>'a'</td>
<td>'pie'</td>
<td>'psolid'</td>
<td>300</td>
<td>1</td>
<td>'red'</td>
</tr>
<tr>
<td>San Antonio</td>
<td>0.024</td>
<td>-0.030</td>
<td>2</td>
<td>2</td>
<td>3</td>
<td>'a'</td>
<td>'label'</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
```

And we can create the map using the same code as before:

```plaintext
title1 ls=1.5 "Some Cities in Texas";
pattern1 v=s c=white;
proc gmap map=my_map data=my_map anno=anno_cities;
id statecode;
choro segment / levels=1 nolegend coutline=gray88;
run;
```
The following are a few examples of SAS maps with annotated markers at specific coordinates. Some are simple, and some are a bit fancier …
SAS/GRAPH: The Basics

Classic Rock Stations
Top 10 Lists!

Hover over markers to see lists

Data Source: Estimated from DOE/EIA Electric Power Monthly
Roll over a state to view a list of its reactors

More than 50%
25% to 50%
1% to 24%
None

*Imitation/Enhancement of map from PBS.org website.
*Just a proof-of-concept -- this information is not guaranteed complete & accurate.
Annotating Bubble Markers

What if you want to represent some response data (such as population or sales) at the specified latitude/longitude locations? Instead of using a simple point marker, you might consider using bubbles of varying size, and maybe varying color. The technique is essentially the same as point markers (see example above), but with the size and/or color variables in the annotate dataset controlled by values in the data.

Of course, I don’t want you to get bored, therefore instead of using the exact same example, let’s use something new where I can also teach you a few other tricks along the way. Instead of using mapsgefk.uscity (which already has the latitude/longitude), let’s just start with a list of cities and their population:

```sas
data anno_pies;
  format population_2013 comma12.0;
  input population_2013 city $ 9-50;
  statecode='TX';
  anno_flag=1;
  datalines;
  1258000 Dallas
  2196000 Houston
  1409000 San Antonio
  100223 Tyler
; run;
```

We can now use the cool (relatively new) Proc Geocode to estimate the latitude and longitude centroid of each city. Note that I used the name ‘statecode’ instead of ‘state’ in my data, so that it will match my map data. And similarly, I rename ‘x’ and ‘y’ to ‘long’ and ‘lat’ to match the variables used in the map dataset.
proc geocode data=anno_pies
   out=anno_pies (rename=(x=long y=lat))
   addressstatevar=statecode method=city;
run;

Now that we have the lat/long centroids, we can combine the response data with the map, gproject them, and separate them again (similar to the previous example):

data my_map; set mapsgfk.us_counties
   (where=(statecode='TX' and density<=1));
run;

data combined; set my_map anno_pies; run;
proc gproject data=combined out=combined
   eastlong degrees latlong dupok;
   id statecode county;
run;

data my_map anno_pies; set combined;
if anno_flag=1 then output anno_pies;
else output my_map;
run;
We now need to add a few variables to the dataset, so that it can be interpreted as annotate commands. This is the same as was explained in the previous example, except this time the size is based on the population, and the color is based on the city name. The size is the radius of the pie, and I want calculate a radius (size) that makes the area of the bubbles proportional to their population. I use the following equation (where the .007 is just a scaling factor): size=sqrt(population_2013/3.14)*.007. Note that there are two observations for each bubble – one to draw the solid filled bubble, and one to draw a gray outline around it.

```sas
data anno_pies; set anno_pies;
xsys='2'; ysys='2'; hsys='3'; when='a';
length function style $8 color $20;
function='pie'; rotate=360; style='psolid';
run;

data anno_pie1; set anno_pies;
    size=sqrt(population_2013/3.14)*.007;
    if city='Dallas' then color='cx1464F4';
    else color='cxFFFF00'; output;
    style='pempty'; color='gray33'; output;
run;
```

Population in TX Cities
The resulting map looks pretty good – there is a bubble marker at each city, with the size is proportional to the population and the color controlled by the city (let’s say, for example, we want Dallas to be blue because it’s the distribution hub for our imaginary company). But what if we wanted more visual ‘impact’ from the bubbles, and therefore we wanted them to be bigger? That’s easy – we can just increase the scaling factor (currently .007) in the size equation: \( \text{size} = \sqrt{\frac{\text{population}_{2013}}{3.14}} \times 0.007 \). Let’s increase it to .02 …

```sas
data anno_pie2; set anno_pies;
  size=sqrt(population_2013/3.14)*.02;
  if city='Dallas' then color='cx1464F4';
  else color='cxFFFF00'; output;
  style='pempty'; color='gray33'; output;
run;
```

Larger bubbles can be problematic though – they obscure the map, and could even obscure other bubbles. In situations like this, my favorite work-around is to use transparent colors to fill the bubbles. Whereas you specify a regular RGB color using `cxRRGGBB`, you specify a transparent color as `aRRGGBBnn` (where `nn` is a hexadecimal number that controls the amount of transparency, with a low `nn` being more transparent, and a high `nn` being more solid). In this case, instead of using `cxFFFFF00` (yellow), simply use `aFFFF0077` (transparent yellow), then you can better see the state borders and overlapping bubbles:
data anno_pie3; set anno_pies;
size=sqrt(population_2013/3.14)*.02;
if city='Dallas' then color='a146F4aa';
else color='aFFFF0077';
output;
style='pempty'; color='gray33'; output;
run;

And there you have it – large bubbles, that don’t obscure the map or the other bubbles!
The following are some real-world examples of maps with bubble markers, of varying sizes, annotated at specific locations on maps …

### Murder Rate in Major U.S. Cities

Data Source: 2009 data from census.gov

### Olympic Athletes per Country
Chapter 7: Markers on Maps

DirecTV-4S Coverage Patterns

This is just a proof-of-concept -- do not consider as accurate!

U.S. Upland Cotton Quality
2003-04 Crop, as of December 11, 2003
Annotating Bubbles at Map Polygon Centroids

With a traditional choropleth map, you can represent the amount (or magnitude) of the data values with the colors or shades of the areas in the map. For example, let’s create a map showing the population of the states in the northeast US, from the sashelp.us_data dataset. Below is the code that subsets the data, and plots it on a simple choro map:

data my_data;
set sashelp.us_data (where=(region='Northeast'));
run;

title1 ls=1.5 "Northeast US Population";
legend1 label=none across=1 position=(top left) mode=share
   order=descending shape=bar(.15in,.15in) offset=(8,-8);
proc gmap map=maps.us data=my_data;
id statecode;
choro population_2010 / coutline=graycc legend=legend1;
run;
A choro map is often a good way to plot data on a map, but there are several limitations. Since not all map areas are the same size, it is often difficult to easily compare the values. For example, what if a very small state has a very large population? … The state would be in the brightest color, but it wouldn’t make a very big visual impact because of the small size of the state. One alternative that makes it easier to compare the values is to use bubbles to represent the data values. Then the varying physical sizes of the states are not an issue, because you’re visually comparing the sizes of the bubbles.

Unlike the gmap block map (which plots a bar in the middle of each map polygon – in this case states), there is no built-in option to plot a bubble in the middle of each state. We have to calculate the x/y location where we want the bubble, and then annotate a pie (bubble). The following code subsets the map to get just the states we want, and then uses the SAS `%centroid` macro to estimate the x/y center of each state:

```
data my_map; set maps.us (where=(statecode in
   ('CT' 'ME' 'MA' 'NH' 'NJ' 'NY' 'PA' 'RI' 'VT'))));
run;
%annomac;
%centroid(my_map,state_centroids,state);
```

Now that we’ve got the centroid for each state, let’s merge that data with our response data (in this case the population), and then we’ll have the basis for our annotate dataset (anno_bubbles). There are several ways to merge data in SAS – my favorite is using SQL. Note that you don’t need to have the numeric state fips code (state) or the 2-character state abbreviation (statecode), but I like to include such variables because they make it easier to verify the results …

```
proc sql;
create table anno_bubbles as
select unique state_centroids.*,
    my_data.statecode, my_data.population_2010
from state_centroids left join my_data
on state_centroids.state=my_data.state
```
order by population_2010 descending;
quit; run;

Now it's just a simple matter of adding a few more variables to the dataset so that it can be used as an annotate dataset to draw the bubbles. The variables are pretty much the same as the ones in the previous example. Here's the data step code, followed by the first few observations of the dataset:

data anno_bubbles; set anno_bubbles;
length color $9;
xsys='2'; ysys='2'; hsys='3'; when='a';
function='pie'; rotate=360;
size=sqrt(population_2010/3.14)*.003;
style='psolid'; color='cxEE2C2C'; output;
style='pempty'; color='grayaa'; output;
run;

And now it's a simple matter of plotting a blank choro map, and annotating the bubble markers on it!
title l=1.5 "Northeast US Population"
pattern v=s c=white;
proc gmap map=my_map data=my_map anno=anno_bubbles;
id statecode;
choro segment / levels=1 nolegend coutline=gray88;
run;

Northeast US Population

And, similar to the previous example, you can use bigger bubbles (multiply the size by .008 instead of .003), and a transparent color (aEE2C2C99 instead of cxEE2C2C) to get a slightly different look …
Here are a couple of real-world examples, where markers were annotated at the centroids of the map areas …
Chapter 7: Markers on Maps

US Immigrant EB5 Visas, FY-2014

Clothing Sales in Department Stores
Year = 1990

Data Source: Sales & Marketing Mgmt. Magazine
NCSU Textile/Apparel Business Information System (TABS)
Annotating Paths

Sometimes you don’t want to annotate individual markers on a map, but rather a series of connected locations. Perhaps you want a line representing locations where a person, vehicle, or animal have been. Or perhaps you’re plotting a highway or railroad tracks on a map. This is another situation where annotate comes in handy.

All you need is some latitude/longitude points along the path, and then you can connect those points with an annotated line.

For this example, I’ll use a simplified version of the proposed Atlantic Coast Pipeline route. I’m only including a few of the data points here, to make it easier for you to copy-n-paste (if you download my code from the samples page, you’ll get extra points along the route).

data anno_route;
  input lat long;
  flag=1;
  datalines;
  39.17 -80.57
  39.15 -80.55
  39.14 -80.45
  39.10 -80.41
  39.02 -80.32
  38.83 -80.11
  38.50 -79.68
  38.27 -79.41
  38.21 -79.16
  37.91 -78.80
  37.41 -78.42
  37.13 -77.96
  36.92 -77.86
  36.47 -77.55
  35.62 -78.12
  35.19 -78.68
  34.74 -79.19
;
run;

Here’s what the first few lines of data look like:
Now you’ll want to get the map, combine it with the point data, and then project them. In this case, since the pipeline will be going through three states, we’ll want those three states (NC, VA, and WV) in our map. Let’s subset the map based on the density values, so that fewer points will be used, and the borders will look slightly less busy.

```sas
data my_map; set mapsgfk.us_counties
  (where=(statecode in ('NC' 'VA' 'WV')
    and density<=2) drop=resolution);
run;

data combined; set my_map anno_route;
run;
proc gproject data=combined out=combined
  latlong degrees eastlong dupok;
id statecode county;
run;
data my_map anno_route; set combined;
if flag=1 then output anno_route;
else output my_map;
run;
```

Now that the lat/long coordinates in our map and our annotated route have been projected into X/Y coordinates that will look good on a flat page, let’s add some annotate commands to tell SAS what to do with the X/Y points in the annotated route dataset. First, so you can ‘see’ where the points will show up on the map, let’s plot them as red dots, using the annotate pie function.
data anno_route; set anno_route;
length function $8 color $8 style $35;
xsys='2'; ysys='2'; hsys='3'; when='a';
function='pie'; style='psolid'; rotate=360;
color='red'; size=.5;
run;

And now we can easily plot a simple/blank map, and specify the above data using the anno= option, and get the following map:

pattern1 v=s c=white;
proc gmap data=my_map map=my_map all anno=anno_route;
id statecode county;
choro segment / levels=1 nolegend coutline=grayd0;
run;
But what we really want to do is connect the points with a line, therefore let’s modify the annotate dataset and use the annotate move and draw functions, rather than the pie function. To connect the points with a line, you’ll want to move to the first observation (which is at the beginning of the route), and then you’ll want to draw to each of the following points.

```sas
data anno_route; set anno_route;
length function $8 color $8;
xsys='2'; ysys='2'; hsys='3'; when='a';
if _n_=1 then function='move';
else function='draw';
color='red'; size=.5; style=''; rotate=.;
run;
```
proc gmap data=my_map map=my_map all anno=anno_route;
id statecode county;
choro segment / levels=1 nolegend coutline=grayd0;
run;
And now that we’ve got the line annotated on the map, let’s annotate some title text (we’re annotating, rather than using the built-in title statements, so that the titles can share the map space instead of being at the top of the page).

data anno_text;
length function $8 color $8 style $35 text $100;
xsys='3'; ysys='3'; hsys='3'; when='a';
function='label'; position='5'; style='albany amt/bold';

color="red"; size=4.7;
x=20; y=92; text="Atlantic Coast"; output;
y=y-5.5; text="Pipeline"; output;

color="grayaa"; size=3.5;
x=70; y=90; text="Proposed Route"; output;

run;

And now we can produce the final map! …

proc gmap data=my_map map=my_map all anno=anno_route;
id statecode county;
choro segment / levels=1 nolegend coutline=grayd0
  anno=anno_text;
run;
And now, here are a few real-world examples of paths annotated on a map …

Tracking Sea Turtles
Mouse over markers to see detailed data.
Click on markers to see Google map.
Chapter 7: Markers on Maps

Proc OptNet: MinSpanTree
(connecting the state capitals)
Search for Missing Flight MH370

For more context regarding this map please visit the SAS blog

Napoleon’s Russian Campaign, 1812

Plotted on modern map
Annotating Borders

SAS/Graph Proc GMap is only designed to represent 1 level of borders on a map. For example, if you’re plotting a county map, GMap only shows the county borders (not the state borders). Therefore if you want to show extra levels of borders, you’ll need to do a little extra work. As with most customizations, we can use annotate to add the extra borders.

Let’s start with a county map of North and South Carolina. Since we’re subsetting this out of the US map, we’ll need to project it (otherwise it will look ‘crooked’ since it was projected based on the center of the US rather than the center of the Carolinas).

data my_map; set mapsgfk.us_counties (where=(statecode in ('NC' 'SC') and density<=2) drop=resolution);
run;

proc gproject data=my_map out=my_map
  latlong eastlong degrees;
id state county;
run;

title1 ls=1.5 "North and South Carolina";
pattern1 v=s c=white;
proc gmap map=my_map data=my_map;
id state county;
choro segment / levels=1 nolegend coutline=grayaa;
run;

North and South Carolina
There are two states in this map, but it is difficult to quickly tell where one stops and the other starts, without the state borders shown. The first step to annotating the state borders is to create an actual map of the state borders, using Proc GRemove. We could hypothetically use the state map, but we would have to take special care to project it in exactly the same way the county map was projected … and even then there’s no guarantee that the state map would be showing the exact same areas as the county map (especially when it comes to showing river & coastal borders, islands, etc). Therefore I always like to use the exact same map, and GRemove the unwanted internal borders – this guarantees the borders in the state map will match the county map! (You don’t really need to plot this intermediate map, but I’m plotting it so you can see that it is indeed a map dataset.)

```
proc gremove data=my_map out=anno_outline;
   by state notsorted;
   id county;
run;

proc gmap map=anno_outli
ne data=anno_outline;
   id state;
   choro segment / levels=1 nolegend coutline=grayaa;
run;
```

North and South Carolina

Proc GRemove Internal Boundaries
And now the final step is to convert the state outline map dataset into an annotate dataset that will simply draw the state outlines. The first observation of each state will become the annotate ‘poly’ function, and all subsequent observations will be ‘polycont’. Poly tells annotate to start drawing a new polygon, and polycont tells annotate to continue drawing the polygon. When annotate encounters the last polycont observation, it then connects that point back to the first one. You might be asking why we didn’t use “by state” – we could do that in most cases, but since some states could have multiple segments, it’s best to use “by state segment”. This code can be re-used with most any map that has a compound id, by simply changing the ‘state’ and ‘county’ variables to the variables used in your map.

```sas
data anno_outline; set anno_outline;
by state segment notsorted;
length function $8 color $8;
color='gray33'; style='mempty'; when='a'; xsys='2'; ysys='2';
if first.segment then function='poly';
else function='polycont';
run;
```

Here are the first few observations in the anno_outline dataset (notice that I’m making the annotated state outline a little darker than the county borders):

<table>
<thead>
<tr>
<th>xsys</th>
<th>ysys</th>
<th>X</th>
<th>Y</th>
<th>function</th>
<th>color</th>
<th>style</th>
<th>when</th>
</tr>
</thead>
<tbody>
<tr>
<td>2</td>
<td>2</td>
<td>-0.083323</td>
<td>0.013196</td>
<td>poly</td>
<td>gray33</td>
<td>mempty</td>
<td>a</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>-0.060575</td>
<td>0.013084</td>
<td>polycont</td>
<td>gray33</td>
<td>mempty</td>
<td>a</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>-0.058803</td>
<td>0.012985</td>
<td>polycont</td>
<td>gray33</td>
<td>mempty</td>
<td>a</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>-0.057821</td>
<td>0.012951</td>
<td>polycont</td>
<td>gray33</td>
<td>mempty</td>
<td>a</td>
</tr>
<tr>
<td>2</td>
<td>2</td>
<td>-0.052285</td>
<td>0.012828</td>
<td>polycont</td>
<td>gray33</td>
<td>mempty</td>
<td>a</td>
</tr>
</tbody>
</table>
And now we can annotate the state outlines on the county map, using the following code:

```sas
proc gmap map=my_map data=my_map anno=anno_outline;
  id state county;
  choro segment / levels=1 nolegend coutline=graydd;
run;
```

North and South Carolina

On the following pages are some examples of fancier maps created using this technique. Note that in the above simple example I’m not actually plotting any response data on the map (all the map areas are the same color), but of course you can also use this technique on a colored choro map where you’re plotting response data.
Space Shuttle Debris Field
Mouse over counties to see debris count data. Click on green counties to see 'mapquest' roadmap of that area. Click on blue counties to see detailed map of debris in that county.

Who Makes Your Girl Scout Cookies?
Predominant Church Bodies by County

SO₄ Concentration (estimated using Loess)

Plotted using SAS/Graph PROC GMAP and ODS HTML Overlay
County map, with annotated state outlines

Mouse over areas to see county name and SO₄ value
Click on states to go to their EPA page
Outbreak of Ebola Virus Disease (EVD) in West Africa

- **Cases in past 21 days**
  - Newly Infected - New cases in previous 7 days
  - Active - New Cases in previous 21 days
  - Not Active - No cases in previous 21 days
  - Major Ebola Treatment Center (ETC) HUBS

Proof-of-Concept Using Estimated Data